
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of EMM test case 9.3.1.15 which are part of the LTE test suite.

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_9_3_1_15
Test Group:
Idle mode

ATS Version:
iwd-EUTRA-TVB2009-12_D11wk04
System Simulator used:
Anite Conformance Toolset
UE(s) used:                       ICERA Espresso 410
Verification Status:
PASS
4 Corrections required for test case 9.3.1.15

4.1 Introduction

This section describes the changes required to make test case 9.3.1.15 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-12_D10wk04 release.

4.2 Change 1

Testcase name
9.3.1.15

Reason for change
As per prose R5-110751, this test case is only applicable for CS fall back and hence test purpose 1 from prose has been removed. Based on that the sections in TTCN related to service request should be removed. 

Summary of change
Removed the code corresponding to prose step 1 till prose step 16 in TTCN  

Source of change
NAS_ServiceRequest_EOnly.ttcn

Before:

function fl_TC_9_3_1_15_Body() runs on EUTRA_PTC

{

  var GutiParameters_Type v_GutiParams_A := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

  var GutiParameters_Type v_GutiParams_B := f_EUTRA_CellInfo_GetGuti(eutra_CellB);

  var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

  var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams_A.PLMN_Identity);

  var SRB_COMMON_IND v_ReceivedAsp;

  var NAS_MSG_Indication_Type v_NasInd;

  var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

  var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

  var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType();

  timer t_T3417 := f_EUTRA_SetTimerToleranceMin( eutra_CellA, nasTimer, 5.0 );

  timer t_T3417ext := f_EUTRA_SetTimerToleranceMin( eutra_CellA, nasTimer, 10.0 );

  timer t_WaitTimer := f_EUTRA_SetTimerToleranceMax( eutra_CellB, nasTTCNTimer, 5.0 );

  //+   The SS pages the UE using S-TMSI with CN domain indicator set to "PS".

  //@siclog "Step 1" siclog@

  f_EUTRA_UE_Page ( eutra_CellA,

                    cs_NasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParams_A.MME_Code,

                                                                       v_GutiParams_A.M_TMSI)));

  //+   The UE transmits a SERVICE REQUEST message. --> SERVICE REQUEST

  //@siclog "Step 2" siclog@

  v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                             tsc_RRC_TI_Def,

                                             cr_EstablishmentCause_mt_Access,

                                             cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                               cr_508_SERVICE_REQUEST(v_SecurityParams.KSIasme)));

  t_T3417.start;

  //+ The SS does not perform a radio bearer establishment procedure.Note: e.g. SS does not send any AS or NAS messages to UE.

  //@siclog "Step 3" siclog@

  //+ Set the cell type of cell A to the ''Non-Suitable cell''.Set the cell type of cell B to the ''Serving cell''.

  // Note: cell type should be set before timer T3417 expired.

  //@siclog "Step 4" siclog@

  f_EUTRA_SetCellPower ( eutra_CellB, tsc_ServingCellRS_EPRE, cs_TimingInfo_Now );

  f_EUTRA_SetCellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE, cs_TimingInfo_Now );

  if( t_T3417.running)

  {

    t_T3417.stop

    }// else timeout would be caught in the default

  //+  UE initiates RRC connection establishment procedure on cell B.

  //@siclog "Step 5" siclog@

   SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellB,

                                    cr_RRCConnectionRequest(cr_EstablishmentCause_Any)));

   f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellB, cs_TimingInfo_Now, tsc_RRC_TI_Def);

  //+   Check: Does the UE transmit TRACKING AREA UPDATE REQUEST message on cell B in the next 5 seconds?

  //@siclog "Step 6" siclog@

      t_WaitTimer.start;

      alt {

          [] SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellB,

                                                cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                                cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                  cdr_TAU_Request_OldGuti_BearerEst(v_UpdateType,

                                                                                                 f_EUTRA_SecurityKSIasme_Get(),

                                                                                                 v_AdditionalUpdateType,

                                                                                                 f_GutiParameters2MobileIdentity(omit, v_GutiParams_A)))))

                         -> value v_ReceivedAsp

           {

                t_WaitTimer.stop;

                f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6: TAU Request on cell B")

           }// timeout would be caught in the default

      }

  //+    The SS transmits a TRACKING AREA UPDATE ACCEPT message.

  //@siclog "Step 7" siclog@

  v_EPS_ContextStatus := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

  SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_CellB,

                                       cs_TimingInfo_Now,

                                       cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                        cs_508_TAU_Accept ( v_UpdateType,

                                                                            f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams_B),

                                                                            cds_TAIListNonConsecutive_tlv( v_PLMN, {'0002'O} ), //TAC-2

                                                                            v_EPS_ContextStatus,

                                                                            f_EUTRA_GetCellLAI(eutra_CellB),

                                                                            f_GetMSId ( ),

                                                                            f_GetAdditionalUpdateResult(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)) ) ) );

  //+  The UE transmits a TRACKING AREA UPDATE COMPLETE message.

  //@siclog "Step 8" siclog@

  SRB.receive ( car_SRB1_NasPdu_IND ( eutra_CellB,

                                          cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                              cr_508_TAU_Complete ) ) );

  //-   EXCEPTION: Steps 9 to 27 describe behaviour that depends on the UE capability. IF pc_CSfallback is true, from Step 9 to step 27should be executed, or else test case is terminated in step 8

  if(pc_CS_Fallback){

  //+   If possible (see ICS) switch off is performed. Otherwise the power is removed.

  //@siclog "Step 9" siclog@

  //+  the UE transmit a DETACH REQUEST with the Detach Type IE indicating "switch off" .( optional)

  //  Note: whether send this message depends on switch-off or power-off.

  //@siclog "Step 10" siclog@

  f_EUTRA_UE_Detach_SwitchOffUe ( eutra_CellB, RRC_CONNECTED );

  //+  The UE is switched on.

  //@siclog "Step 11" siclog@

  f_UT_SwitchOnUE(UT, false);

  //+    The UE transmits an ATTACH REQUEST including a PDN CONNECTIVITY REQUEST message on cell B.

  //-   EXCEPTION: Steps 13a1 to 13a2 describe behaviour that depends on UE configuration; the "lower case letter" identifies a step sequence that take place if the UE has ESM information which needs to be transferred.

  //+  IF the UE sets the ESM information transfer flag in the last PDN CONNECTIVITY REQUEST message THEN the SS transmits an ESM INFORMATION REQUEST message to initiate exchange of protocol configuration options and/or APN.

  //+  The UE transmits an ESM INFORMATION RESPONSE message to transfer protocol configuration options and/or APN.

  //+   SS responds with ATTACH ACCEPT message. The ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message is piggybacked in ATTACH ACCEPT message

  //+   The UE transmits an ATTACH COMPLETE message including an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message

  //+   The SS releases the RRC connection.

  //@siclog "Step 12 - 16" siclog@

  f_EUTRA_AttachProcedure (eutra_CellB, "");
  //+  Force the UE to initiate CS Voice call. (Note 1)

  //@siclog "Step 17" siclog@

  f_UT_RequestCSCall(UT);

  …

} // function fl_TC_9_3_1_15_Body

After:

function fl_TC_9_3_1_15_Body() runs on EUTRA_PTC

{

  var GutiParameters_Type v_GutiParams_A := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

  var GutiParameters_Type v_GutiParams_B := f_EUTRA_CellInfo_GetGuti(eutra_CellB);

  var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

  var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams_A.PLMN_Identity);

  var SRB_COMMON_IND v_ReceivedAsp;

  var NAS_MSG_Indication_Type v_NasInd;

  var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

  var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

  var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType();

  timer t_T3417 := f_EUTRA_SetTimerToleranceMin( eutra_CellA, nasTimer, 5.0 );

  timer t_T3417ext := f_EUTRA_SetTimerToleranceMin( eutra_CellA, nasTimer, 10.0 );

  timer t_WaitTimer := f_EUTRA_SetTimerToleranceMax( eutra_CellB, nasTTCNTimer, 5.0 );

  var template (value) CellPowerList_Type v_CellPowerList; 

  //Step 1-16 are now void
  //+  Force the UE to initiate CS Voice call. (Note 1)

  //@siclog "Step 17" siclog@

  f_UT_RequestCSCall(UT);

  …

} // function fl_TC_9_3_1_15_Body



4.3 Change 2

Testcase name
9.3.1.15

Reason for change
1) As per prose CR, SS should trigger handover of UE from cellA to cellB for the UE to detect a change in tracking area. Hence UE should be in state 3, generic RB established to make the test case more similar to real network scenario.
2) As mentioned in prose CR R5-110751, the UE would now do preamble on cellA and then do handover to cell B. Hence the postamble should happen on cellB

3) Since UE is in generic RB established mode, it will receive the extended service request on SRB2 on cellA

4) As per CR R5-110751, the SS would do handover of UE to cell B.  Based on that, the power level of cellA should be configured to that of Suitable neighbour cell and there is no need to start the timer T3417 

5) After handover, UE would send TAU request on cellB on SRB2 with GUTI of cell A as mobile identity. 

6) SS should send TAU Accept on SRB2 on cell B with TAC and LAI of cellB

7) SS should receive TAU complete on SRB2 on cellB

8) UE would transmit extended service request on SRB2 on cellB and SS should use the updated security context and GUTI parameters of cellB

9) SS should send the service reject on cell B as integrity protected and ciphered message since security context is established. 

10) SS should release the connection on cellB 



Summary of change
1) UE configured in generic RB established mode after preamble
2) Postamble done on cellB

3) extended service request received on SRB2 on cellA

4) Handover done to cellB

5) TAU request handled on cellB on SRB2 with GUTI of cell A as mobile identity. 

6) Sent TAU Accept on SRB2 on cell B with TAC and LAI of cellB

7) Received TAU complete on SRB2 on cellB

8) Handled extended service request on SRB2 on cellB using the updated security context and GUTI parameters of cellB

9) Sent the service reject on cell B as integrity protected and ciphered message. 

10) Released the connection on cellB 



Source of change
NAS_ServiceRequest_EOnly.ttcn

Before:

function f_TC_9_3_1_15_EUTRA() runs on EUTRA_PTC

{

  …

  var template (value) CellPowerList_Type v_CellPowerList;

  f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c3);

  // create cell and send out system information

  f_EUTRA_CellConfig_Def(eutra_CellA);

  f_EUTRA_CellConfig_Def(eutra_CellB);

  v_CellPowerList := {

    cs_CellPower ( eutra_CellA, tsc_ServingCellRS_EPRE ),

    cs_CellPower ( eutra_CellB, tsc_NonSuitableCellRS_EPRE )

  };

  f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

  f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

  f_EUTRA_TestBody_Set(true);

  fl_TC_9_3_1_15_Body();

  // Postamble

  f_EUTRA_TestBody_Set(false);

  f_EUTRA_Postamble ( eutra_CellA, E1_IDLE );

} // function f_TC_9_3_1_15_EUTRA

function fl_TC_9_3_1_15_Body() runs on EUTRA_PTC

{

  var GutiParameters_Type v_GutiParams_A := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

  var GutiParameters_Type v_GutiParams_B := f_EUTRA_CellInfo_GetGuti(eutra_CellB);

  var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

  var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams_A.PLMN_Identity);

  var SRB_COMMON_IND v_ReceivedAsp;

  var NAS_MSG_Indication_Type v_NasInd;

  var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

  var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

  var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType();

  timer t_T3417 := f_EUTRA_SetTimerToleranceMin( eutra_CellA, nasTimer, 5.0 );

  timer t_T3417ext := f_EUTRA_SetTimerToleranceMin( eutra_CellA, nasTimer, 10.0 );

  timer t_WaitTimer := f_EUTRA_SetTimerToleranceMax( eutra_CellB, nasTTCNTimer, 5.0 );

  …

  //+  Force the UE to initiate CS Voice call. (Note 1)

  //@siclog "Step 17" siclog@

  f_UT_RequestCSCall(UT);

  //+ Step 18   The UE transmits an EXTENDED SERVICE REQUEST message.

  //@siclog "Step 18" siclog@

  v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,

                                             tsc_RRC_TI_Def,

                                             cr_EstablishmentCause_MoData,

                                             cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                               cr_EXT_SERVICE_REQUEST_MO(v_SecurityParams.KSIasme,

                                                                                          bit2oct(v_GutiParams_B.M_TMSI))));

  t_T3417ext.start;

  //+    The SS does not respond to this request. Note: e.g. SS does not send any AS or NAS messages to UE.

  //@siclog "Step 19" siclog@

  //+   Set the cell type of cell A to the '' Serving cell ''.

  //+ Step Set the cell type of cell B to the '' Non-Suitable cell ''.Note: cell type should be set before timer T3417ext expired.

  //@siclog "Step 20" siclog@

  f_EUTRA_SetCellPower ( eutra_CellA, tsc_ServingCellRS_EPRE, cs_TimingInfo_Now );

  f_EUTRA_SetCellPower ( eutra_CellB, tsc_NonSuitableCellRS_EPRE, cs_TimingInfo_Now );

  if( t_T3417ext.running)

  {

    t_T3417ext.stop

    }// else timeout would be caught in the default

  //+   UE initiates RRC connection establishment procedure on cell A.

  //@siclog "Step 21" siclog@

   SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellA,

                                    cr_RRCConnectionRequest(cr_EstablishmentCause_Any)));

   f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellA, cs_TimingInfo_Now, tsc_RRC_TI_Def);
   //+   Check: Does the UE transmit TRACKING AREA UPDATE REQUEST message on cell A in the next 5 seconds?

   //@siclog "Step 22" siclog@

      t_WaitTimer.start;

      alt {

          [] SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                                cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                                cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                   cdr_TAU_Request_OldGuti_BearerEst(v_UpdateType,

                                                                                                 f_EUTRA_SecurityKSIasme_Get(),

                                                                                                 v_AdditionalUpdateType,

                                                                                                 f_GutiParameters2MobileIdentity(omit, v_GutiParams_B)))))

                         -> value v_ReceivedAsp

           {

                t_WaitTimer.stop;

                f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 22: TAU Request on cell A")

           }// timeout would be caught in the default

      }

  //+    The SS transmits a TRACKING AREA UPDATE ACCEPT message.

  //@siclog "Step 23" siclog@

  v_EPS_ContextStatus := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

  SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_CellA,

                                       cs_TimingInfo_Now,

                                       cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                        cs_508_TAU_Accept ( v_UpdateType,

                                                                            f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams_A),

                                                                            cds_TAIListNonConsecutive_tlv( v_PLMN, {'0001'O} ), //TAC-1

                                                                            v_EPS_ContextStatus,

                                                                            f_EUTRA_GetCellLAI(eutra_CellA),

                                                                            f_GetMSId ( ),

                                                                            f_GetAdditionalUpdateResult(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)) ) ) );

  //+   The UE transmits a TRACKING AREA UPDATE COMPLETE message.

  //@siclog "Step 24" siclog@

  SRB.receive ( car_SRB1_NasPdu_IND ( eutra_CellA,

                                          cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                              cr_508_TAU_Complete ) ) );

  //+  Check: Does the UE transmit EXTENDED SERVICE REQUEST message on cell A? [2,P]

  //@siclog "Step 25" siclog@

  SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cr_EXT_SERVICE_REQUEST_MO(v_SecurityParams.KSIasme,

                                                                               bit2oct(v_GutiParams_A.M_TMSI)))));

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 25");

  //+    The SS sends SERVICE REJECT message in order that the UE enters EMM-REGISTERED.NORMAL-SERVICE.

  //@siclog "Step 26" siclog@

  SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellA,

                               cs_TimingInfo_Now,

                               cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                              cs_508_SERVICE_REJECT(tsc_Cause_CSDomainNotAvailable))));

  //+  The SS releases the RRC connection.

  //@siclog "Step 27" siclog@

  f_EUTRA_RRC_ConnectionRelease(eutra_CellA, tsc_RRC_TI_Def, other);

  }

} // function fl_TC_9_3_1_15_Body

After:

function f_TC_9_3_1_15_EUTRA() runs on EUTRA_PTC

{

  …

  var template (value) CellPowerList_Type v_CellPowerList;

  f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c3);

  // create cell and send out system information

  f_EUTRA_CellConfig_Def(eutra_CellA);

  f_EUTRA_CellConfig_Def(eutra_CellB);

  v_CellPowerList := {

    cs_CellPower ( eutra_CellA, tsc_ServingCellRS_EPRE ),

    cs_CellPower ( eutra_CellB, tsc_NonSuitableCellRS_EPRE )

  };

  f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

  f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

  f_EUTRA_RbEst_Def ( eutra_CellA ); 

  f_EUTRA_TestBody_Set(true);

  fl_TC_9_3_1_15_Body();

  // Postamble

  f_EUTRA_TestBody_Set(false);

  f_EUTRA_Postamble ( eutra_CellB, E1_IDLE );

} // function f_TC_9_3_1_15_EUTRA

function fl_TC_9_3_1_15_Body() runs on EUTRA_PTC

{

  var GutiParameters_Type v_GutiParams_A := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

  var GutiParameters_Type v_GutiParams_B := f_EUTRA_CellInfo_GetGuti(eutra_CellB);

  var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

  var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams_A.PLMN_Identity);

  var SRB_COMMON_IND v_ReceivedAsp;

  var NAS_MSG_Indication_Type v_NasInd;

  var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

  var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

  var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType();

  timer t_T3417 := f_EUTRA_SetTimerToleranceMin( eutra_CellA, nasTimer, 5.0 );

  timer t_T3417ext := f_EUTRA_SetTimerToleranceMin( eutra_CellA, nasTimer, 10.0 );

  timer t_WaitTimer := f_EUTRA_SetTimerToleranceMax( eutra_CellB, nasTTCNTimer, 5.0 );

  var template (value) CellPowerList_Type v_CellPowerList; 

  //+  Force the UE to initiate CS Voice call. (Note 1)

  //@siclog "Step 17" siclog@

  f_UT_RequestCSCall(UT);

  SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                               cr_EXT_SERVICE_REQUEST_MO(f_EUTRA_SecurityKSIasme_Get(), 

                                                                                          bit2oct(v_GutiParams_A.M_TMSI)))));

  v_CellPowerList := {

    cs_CellPower ( eutra_CellA, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE ),

    cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE )

  };

  f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );
  //SS now creates a handover situation to direct the UE to cellB

  f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_CellA, eutra_CellB );
  SRB.receive( car_SRB2_NasPdu_IND(eutra_CellB,

                                   cr_NAS_Indication(

                                   tsc_SHT_IntegrityProtected,
                                   cdr_TAU_Request_OldGuti_BearerEst(v_UpdateType,

                                                                     f_EUTRA_SecurityKSIasme_Get(),

                                                                     v_AdditionalUpdateType,

                                                                     f_GutiParameters2MobileIdentity(omit, v_GutiParams_A))))) -> value v_ReceivedAsp;

  v_NasInd := v_ReceivedAsp.Signalling.Nas[0] ;
  v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 22: TAU Request sent by UE");

  // The SS transmits a TRACKING AREA UPDATE ACCEPT message.

  SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_CellB,

                                       cs_TimingInfo_Now,

                                       cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                        cs_508_TAU_Accept ( v_UpdateType,

                                                                            f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams_B),

                                                                            cds_TAIListNonConsecutive_tlv( v_PLMN, {'0002'O} ), //TAC-1

                                                                            v_EPS_ContextStatus,

                                                                            f_EUTRA_GetCellLAI(eutra_CellB),

                                                                            f_GetMSId ( ),

                                                                            f_GetAdditionalUpdateResult(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)) ) ) );

  // Check: Does the UE transmit a TRACKING AREA UPDATE COMPLETE message?

  //@siclog "Step 7" siclog@

  SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cr_508_TAU_Complete) ));

  //+  Check: Does the UE transmit EXTENDED SERVICE REQUEST message on cell A? [2,P]

  //@siclog "Step 25" siclog@

  SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cr_EXT_SERVICE_REQUEST_MO(f_EUTRA_SecurityKSIasme_Get(), 

                                                                               bit2oct(v_GutiParams_B.M_TMSI)))));

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 25");

  //+    The SS sends SERVICE REJECT message in order that the UE enters EMM-REGISTERED.NORMAL-SERVICE.

  //@siclog "Step 26" siclog@

  SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellB,

                               cs_TimingInfo_Now,

                               cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                              cs_508_SERVICE_REJECT(tsc_Cause_CSDomainNotAvailable))));

  //+  The SS releases the RRC connection.

  //@siclog "Step 27" siclog@

  f_EUTRA_RRC_ConnectionRelease(eutra_CellB, tsc_RRC_TI_Def, other);

} // function fl_TC_9_3_1_15_Body



5 Execution Log Files.

5.1 ICERA Espresso E410

The ICERA Espresso E410 UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_9_3_1_15_Icera_log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

[1]
R5s110100: This archive comprises html format execution log file.

